Actualisation du Tableau des Entrées Intermédiaires (TEI) par la méthode de Minimisation de l'Entropie Croisée (MEC)

Amar Seck[1]

Placé au cœur du Tableau des Entrées et Sorties (TES), le Tableau des Entrées Intermédiaires (TEI) est un outil de synthèse précieux pour l'analyse du processus productif au sein d'une économie. Son élaboration ne se fait pas de la même façon pour toutes les années. Pour l'année de base, priorité est donnée aux estimations en niveau. A cette occasion, il est rassemblé une information statistique considérable, qu'il n'est pas possible de mobiliser chaque année à l'identique. Pour les autres années, les comptables nationaux, en enregistrant des données supplémentaires, procèdent à une actualisation du TEI de l'année de référence. La méthode la plus utilisée pour ce cas d'espèce est la méthode Racking-Ration (RAS) qui permet d'équilibrer une matrice lorsqu'il y a des divergences entre les différentes sources de données utilisées. Cet article propose une actualisation du TEI avec la méthode de minimisation de l'Entropie Croisée (MEC). Cette méthode appliquée jusqu'ici sur les Matrices de Comptabilité Sociales (MCS) est très flexible. Elle permet d'incorporer des variables d'erreurs et de fixer les contraintes liées aux nouvelles informations obtenues sur la structure interne et les totaux marginaux d'une matrice. Pour illustrer l'applicabilité de la méthode, une expérimentation est faite sur les comptes nationaux du Sénégal.

Introduction

Le Tableau des Entrées Intermédiaires (TEI) est une pièce indispensable à la construction et à la validation des comptes nationaux. Placé au cœur du Tableau des Entrées-Sorties (TES), sa détermination assure la cohérence des trois approches permettant de reconstituer le Produit Intérieur Brut (PIB) : l'approche demande (somme des emplois finaux hors importations), l'approche production (solde entre la production et la Consommation Intermédiaire (CI) d'une branche) et l'approche revenu (calcul du PIB à partir des comptes des secteurs institutionnels par la mesure « directe » de leur valeur ajoutée).

Par ailleurs, le TEI est un outil de synthèse précieux pour analyser le processus productif national et ses évolutions. Pour suivre ces évolutions, les TEI sont établis en valeur, volume et prix chaînés (c'est-à-dire aux prix de l'année précédente, chaînés, année de base).

Quant à l'élaboration du TEI, elle ne se fait pas de la même façon pour toutes les années. Pour l'année de base, priorité est donnée aux estimations en niveau. A cette occasion, on rassemble une information statistique considérable, qu'il n'est pas possible de mobiliser chaque année à l'identique. Pour les autres années, les comptables nationaux, en enregistrant des données supplémentaires, procèdent le plus souvent à une actualisation du TEI de l'année de référence. Ainsi, beaucoup de méthodes sont élaborées pour résoudre ce problème. Celles-ci ont dans leur principe général, l'idée suivante : *« partir d'un TEI de base de*

[1] *Amar Seck, Statisticien, Agence Nationale de la Statistique et de la Démographie - Sénégal* Amar.seck@ansd.sn

l'année (n) obtenu à partir d'enquêtes lourdes et coûteuses sur des délais relativement longs, et, moyennant certaines hypothèses de travail et données comptables annuelles (comptes de branches et équilibres ressources emplois), estimer des TEI annuels pour les années (n+1) et suivantes ». Parmi ces méthodes, la plus utilisée est la méthode RAS[2]. Cependant, dans certains cas, cette méthode n'est pas satisfaisante car elle ne permet pas d'intégrer facilement les nouvelles informations recueillies sur la structure des CI.

Pour une amélioration de cette méthode nous proposons dans cet article la méthode de minimisation de l'Entropie-Croisée pour produire le TEI d'une année courante en partant principalement de celui d'une année de référence. Cette méthode à l'avantage d'être très flexible, car permettant d'incorporer des variables d'erreurs et de fixer les contraintes liées aux nouvelles données collectées.

La première section de cet article fait une présentation de la structure du TEI et une description mathématique du problème d'estimation. Ensuite, nous présentons, en deuxième section, la méthode RAS en relatant ces avantages et inconvénients. En troisième partie, nous décrivons la méthode de la minimisation de l'Entropie Croisée (MEC) et son application au TEI avec une illustration utilisant les comptes nationaux du Sénégal.

La structure du Tableau des Entrées Intermédiaires(TEI)

Le Tableau des Entrées Intermédiaires (TEI) contient l'ensemble des biens et services qui entrent dans le processus de production d'autres biens et services. Il constitue la partie centrale du Tableau des Entrées-Sorties (TES). Il s'agit d'un tableau carré comprenant autant de lignes que de colonnes. Chaque colonne ou ligne correspond à une branche d'activité ou à un produit. Une ligne décrit la ventilation de la Consommation Intermédiaire (CI) totale d'un produit par les différentes branches utilisatrices. Une colonne représente une décomposition de la CI totale d'une branche selon les différents produits.

[2] *Racking-Ratio ou Iterative Proportional Fitting*

Représentation schématique du TEI

	Branches d'activités	
Produits	$B_1 \quad \cdots \quad B_j \quad \cdots \quad B_n$	TOTAL
P_1		$y_1 = \sum_j^n x_{1j}$
$\vdots$		$\vdots$
P_i	x_{ij}	$y_i = \sum_j^n x_{ij}$
$\vdots$		$\vdots$
P_n		$y_n = \sum_j^n x_{nj}$
TOTAL	$x_1 = \sum_i^n x_{i1} \ \dots \ x_j = \sum_i^n x_{ij} \ \dots \ x_n = \sum_i^n x_{in}$	$\sum_i^n \sum_j^n x_{ij}$

Soit $\boldsymbol{X}$ une matrice de dimension $\boldsymbol{nxn}$ représentant le TEI. Son terme général $\boldsymbol{x_{ij}}$ donne simultanément la demande de CI de la branche $\boldsymbol{j}$ en $\boldsymbol{i}$ et l'offre de CI du produit $\boldsymbol{i}$ à la branche $\boldsymbol{j}$. On définit :

$$\forall\, \boldsymbol{i}\ \boldsymbol{et}\ \boldsymbol{j}, \qquad \boldsymbol{x_j = \sum_i^n x_{ij}} \quad \boldsymbol{et} \quad \boldsymbol{y_i = \sum_j^n x_{ij}}$$

$\boldsymbol{x_j}$ donne la demande totale de CI de la branche $\boldsymbol{j}$ et $\boldsymbol{y_i}$ l'offre totale de CI du produit $\boldsymbol{i}$.

Ce qui permet de formaliser l'équation $\boxed{OFFRE\ TOTALE\ CI = DEMANDE\ TOTALE\ CI}$ par :

$$\boldsymbol{\sum_j^n x_j = \sum_i^n y_i = \sum_i^n \sum_j^n x_{ij}}$$

Le problème de l'actualisation du TEI

La confection d'un TEI de base pose des problèmes théoriques et pratiques parfois inextricables et nécessite la mobilisation de ressources humaines et financières importantes en plus des délais qui peuvent aller jusqu'à trois ou quatre années de travail laborieux. En effet, elle exige la mobilisation de sources d'envergures et onéreuses telles que les enquêtes de structures, les enquêtes sur les dépenses de consommation des ménages, l'exploitation des sources administratives, etc.

D'autre part, la plupart des TEI de base ne sont disponibles qu'après plusieurs années de clôture de l'exercice concerné et de ce fait, perdent un peu de leur pertinence et utilité.

Ce qui pousse à s'interroger sur la possibilité, en dehors des années de base, de disposer dans des délais raisonnables de TEI annuels, fiables à partir des statistiques généralement disponibles (ERE et comptes de branches).

Face à cette situation, plusieurs pays optent d'élaborer des TEI de base avec un niveau élevé de fiabilité moyennant des sources lourdes disponibles selon une périodicité donnée (quinquennale ou décennale lors du changement d'année de base) et de les actualiser par la suite. Ce sui consiste à déterminer une nouvelle matrice $\boldsymbol{X}' = \left(\boldsymbol{x}'_{ij}\right)_{i,j\in\{1,\dots,n\}}$ du TEI en se basant sur celui de base $\boldsymbol{X}$ et des nouvelles informations collectées notamment la demande et l'offre de Consommation Intermédiaire respectivement par branche d'activités et par produit. En outre, de nouvelles informations clés, relevées sur la structure de l'économie entre deux années de base, peuvent être intégrées avec la fixation et le remplissage de cases qui étaient vides.

La méthode RAS

La méthode RAS est fréquemment utilisée lors de la construction de Matrices de Comptabilité Sociale et d'actualisation de Tableaux Input-Output (TEI). Elle est principalement appliquée pour équilibrer une matrice lorsqu'il y a des divergences entre les différentes sources de données utilisées. Pour utiliser cette méthode, il suffit de spécifier des totaux de contrôle pour chaque ligne et colonne de la matrice à équilibrer. Ensuite, par itération, les valeurs à l'intérieur de la matrice sont ajustées proportionnellement jusqu'à ce que l'égalité entre la somme des lignes et des colonnes de la matrice et les totaux de contrôle spécifiés soit obtenue.

La méthode consiste, en partant d'une matrice $\boldsymbol{A^0}$ et des nouvelles informations obtenues sur les totaux marginaux, à générer une nouvelle matrice $\boldsymbol{A}$ telle que :

$$\boldsymbol{rA^0s = A}$$

sous la contrainte

$$\sum_{i}^{n} \boldsymbol{r_i} = \sum_{j}^{n} \boldsymbol{s_j}$$

Avec $\boldsymbol{r} = (\boldsymbol{r_i})_{i\in\{1,\dots,n\}}$ et $\boldsymbol{s} = \left(\boldsymbol{s_j}\right)_{j\in\{1,\dots,n\}}$ respectivement les vecteurs totaux en ligne et en colonne de la nouvelle matrice $\boldsymbol{A}$.

Il s'agit de calculer une première matrice $\boldsymbol{A^1}$ dérivée de la matrice d'origine $\boldsymbol{A^0}$ en multipliant chacune de ces cellules par un coefficient de redressement (en ligne ou en colonne) de manière à parvenir à une matrice dont le total des lignes (ou des colonnes) est calé sur l'objectif. Ce coefficient est obtenu par le rapport du total en ligne (ou colonne) de la matrice finale $\boldsymbol{A}$ sur le total en ligne (ou en colonne) de la matrice $\boldsymbol{A^0}$. Si le coefficient est calculé dans un premier temps en ligne, on assiste à un décalage des totaux des lignes avec l'objectif. Ainsi, un processus d'itération démarre en calculant une matrice $\boldsymbol{A^2}$ avec la multiplication des cellules de $\boldsymbol{A^1}$ par le nouveau coefficient donné par le rapport du total en colonne de la matrice $\boldsymbol{A}$ sur le total en colonne de $\boldsymbol{A^1}$. Ce qui donne au bout d'un certain nombre d'itérations la matrice recherchée $\boldsymbol{A}$.

L'avantage de cette méthode est qu'elle est simple d'utilisation et qu'elle requiert peu d'information. En outre, si les totaux de contrôle sont cohérents, la convergence est assurée. Par contre, la méthode souffre de certains inconvénients à savoir le manque de fondements théoriques et la difficulté liée à l'ajout des contraintes ou de l'information supplémentaire au processus, ce qui permettrait l'amélioration de l'estimation. Ce qui justifie le remplacement de cette méthode par la méthode de minimisation de l'Entropie Croisée (MEC).

La méthode de minimisation de l'Entropie Croisée (MEC)

La méthode de minimisation de l'entropie-croisée (MEC) est une méthode générale d'optimisation de type Monte-Carlo[3], combinatoire ou continue. Elle a été conçue à l'origine pour la simulation d'événements rares, où des densités de probabilités très faibles doivent être estimées correctement, par exemple dans l'analyse de la sécurité des réseaux, les modèles de file d'attente, ou l'analyse des performances des systèmes de télécommunication. La méthode MEC peut être appliquée à tout problème d'optimisation combinatoire où les observations sont bruitées comme le problème du voyageur de commerce, le problème d'affectation quadratique, le problème d'alignement de séquences, le problème de la coupure maximale et les problèmes d'allocation de mémoire, tout comme des problèmes d'optimisation continue avec de nombreux extrema locaux.

Le fondement de la méthode MEC est formulé en tant que « second principe d'optimisation de

[3] *Les méthodes de simulation Monte Carlo peuvent être vues comme des méthodes d'approximation, même s'il s'agit d'approximations au sens statistique du terme. Il n'y a pas un consensus absolu sur une définition précise de ce qu'est une technique de type Monte Carlo, mais la description la plus habituelle consiste à dire que les méthodes de ce type se caractérisent par l'utilisation du hasard pour résoudre des problèmes centrés sur un calcul. Elles sont en général applicables à des problèmes de type numérique, ou bien à des problèmes de nature elle-même probabiliste.*

l'entropie » par Kapur et Kesavan (1992), pour les situations où l'on dispose d'information a priori : « De toutes les distributions de probabilité qui satisfont les contraintes imposées, on doit choisir celle qui est la plus proche de la distribution donnée a priori ». Selon les auteurs, ce principe est une généralisation d'un principe d'abord énoncé par Jaynes (1957) comme la traduction opérationnelle de la neutralité scientifique. Le même principe est mis en avant par Golan, Judge et Miller (1996). Ces auteurs, et bien d'autres, soutiennent que les méthodes d'estimation fondées sur la minimisation de l'entropie croisée sont de véritables méthodes d'estimation, d'inspiration bayesienne, et non seulement des techniques pour attribuer des valeurs à des variables.

Dans son approche axiomatique à la théorie de l'information, Theil (1967, p. 5) a montré que, pour un événement aléatoire $\boldsymbol{A}$ de probabilité $\boldsymbol{q}$, la quantité d''information contenue dans un message disant que l'événement $\boldsymbol{A}$ s'est produit est mesurée par $\boldsymbol{log}\left(\frac{1}{q}\right)$. Il s'ensuit que $\boldsymbol{log}\left(\frac{1}{q}\right)$ mesure réciproquement la quantité d'information qui manque ou le degré d'incertitude qui subsiste lorsque l'on ne connaît que la probabilité d'occurrence $\boldsymbol{q}$ de $\boldsymbol{A}$.
Soit maintenant un ensemble exhaustif d'événements mutuellement exclusifs $\{\boldsymbol{A_i}\}$ auxquels sont associées des probabilités $\boldsymbol{p_{ij}}$. Il découle de ce qui précède que l'incertitude que laisse subsister cette distribution de probabilités est mesurée par l'espérance mathématique de la quantité d'information contenue dans un message qui dirait lequel des événements mutuellement exclusifs s'est produit :

$$\boldsymbol{E}^4 = \sum_{i}^{I} \boldsymbol{p_i log}\left(\frac{\boldsymbol{1}}{\boldsymbol{p_i}}\right) = -\sum_{i}^{I} \boldsymbol{p_i log}(\boldsymbol{p_i})$$

Par ailleurs, si l'on veut mesurer la quantité d'information[5] d'un message disant que la probabilité de $\boldsymbol{A}$, que l'on croyait initialement être égale à $\boldsymbol{q}$, est plutôt de $\boldsymbol{p}$, alors il faut la mesurer par la différence entre l'information manquante *ex ante* et l'information manquante *ex post.* Dans ce cas la quantité d'information est donnée par :

$$\boldsymbol{log}\left(\frac{\boldsymbol{1}}{\boldsymbol{q}}\right) - \boldsymbol{log}\left(\frac{\boldsymbol{1}}{\boldsymbol{p}}\right) = \boldsymbol{log}\left(\frac{\boldsymbol{p}}{\boldsymbol{q}}\right)$$

Maintenant, si on suppose qu'un message dit qu'une distribution de probabilités est donnée désormais par la distribution à posteriori $\{\boldsymbol{p_i}\}$, plutôt que par la distribution à priori $\{\boldsymbol{q_i}\}$. Alors la quantité d'information contenue dans ce message est donnée par l'espérance mathématique suivante :

$$\boldsymbol{D}(\boldsymbol{p}:\boldsymbol{q}) = \sum_{i}^{I} \boldsymbol{p_i log}\left(\frac{\boldsymbol{p_i}}{\boldsymbol{q_i}}\right)$$

Cette mesure est appelée, mesure de Kullback-Leibler (1951) pour une distribution uni-dimensionnelle.
Le principe de minimisation de l'entropie croisée consiste donc à minimiser la mesure d'apport d'information de Kullback-Leibler, sous contrainte des identités habituelles respectées par les probabilités, auxquelles peuvent s'ajouter d'autres restrictions propres à chaque situation concrète. Formellement, le problème s'écrit :

$$\boldsymbol{MinD}(\boldsymbol{p}:\boldsymbol{q}) = \min_{(i)} \sum_{i}^{I} \boldsymbol{p_i log}\left(\frac{\boldsymbol{p_i}}{\boldsymbol{q_i}}\right)$$

Sous les contraintes :

$$\boldsymbol{p_i} \geq \boldsymbol{0}, \qquad \sum_{i}^{I} \boldsymbol{p_i} = \boldsymbol{1} \quad \text{et} \quad \sum_{i}^{I} \boldsymbol{p_i g_{ir}} = \boldsymbol{a_r}^{6}$$

L'unicité de la solution de ce problème de minimisation est assurée par la convexité[7] de la fonction objective.

Application de la méthode

En vue de l'application de la méthode de l'entropie croisée au TEI, le problème consiste à minimiser la version bi-dimensionnelle de la mesure d'apport d'information de Kullback-Leibler, sous contrainte des identités habituelles respectées par les probabilités, auxquelles peuvent s'ajouter d'autres restrictions propres à chaque situation concrète. Formellement, le problème s'écrit :

$$\boldsymbol{MinD}(\boldsymbol{p}:\boldsymbol{q}) = \min_{(i,j)} \sum_{j}^{n} \sum_{i}^{n} \boldsymbol{p_{ij} log}\left(\frac{\boldsymbol{p_{ij}}}{\boldsymbol{q_{ij}}}\right) \qquad (\boldsymbol{1})$$

Sous les contraintes :

$$\begin{cases} \boldsymbol{p_{ij}} \geq \boldsymbol{0} & (\boldsymbol{2}) \\ \sum_{j}^{n} \sum_{i}^{n} \boldsymbol{p_{ij}} = \boldsymbol{1} & (\boldsymbol{3}) \\ \sum_{j}^{n} \sum_{i}^{n} \boldsymbol{p_{ij} g_{ijr}} = \boldsymbol{a_r} & (\boldsymbol{4}) \end{cases}$$

Ce bloc d'équation de contraintes permet d'intégrer les nouvelles informations relatives à l'offre et la demande de CI ainsi qu'à l'intérieur de la matrice.

[4] Mesure d'entropie de Shanon
[5] Définition axiomatique de la théorie de l'information

[6] Cette dernière relation permet d'intégrer les contraintes spécifiques au problème d'estimation
[7] Lorsque la fonction objectif d'un problème d'optimisation est convexe, les conditions de premier ordre suffisent à définir un minimum absolu

Pour la mise en pratique, les flux du TEI sont transformés en probabilités. La distribution de probabilité a priori, $\boldsymbol{Q} = (\boldsymbol{q}_{i,j})_{i,j\in\{1,\dots,n\}}$ est donnée par :

$$\boldsymbol{q}_{ij} = \frac{\boldsymbol{x}_{ij}}{\sum_i^n \sum_j^n \boldsymbol{x}_{ij}} \qquad (5)$$

Soient $\boldsymbol{y'}_i$ *et* $\boldsymbol{x'}_j$ respectivement l'offre et la demande de CI, on définit la matrice de distribution de probabilité à posteriori $\boldsymbol{P} = (\boldsymbol{p}_{i,j})_{i,j\in\{1,\dots,n\}}$, par :

$$\begin{cases} \boldsymbol{p}_i = \dfrac{\boldsymbol{y'}_i}{\sum_i^n \boldsymbol{y'}_i} & (6) \\ \boldsymbol{p}_j = \dfrac{\boldsymbol{x'}_j}{\sum_j^n \boldsymbol{x'}_j} & (7) \\ \displaystyle\sum_j^n \boldsymbol{p}_{ij} = \boldsymbol{p}_i & (8) \\ \displaystyle\sum_i^n \boldsymbol{p}_{ij} = \boldsymbol{p}_j & (9) \\ \displaystyle\sum_j^n \sum_i^n \boldsymbol{p}_{ij} = \boldsymbol{1} & (10) \end{cases}$$

Pour le problème d'optimisation donné par la relation $(\boldsymbol{1})$, on remplace le bloc des contraintes $[(\boldsymbol{2})$ à $(\boldsymbol{4})]$ par le bloc d'équations $[(\boldsymbol{8})$ à $(\boldsymbol{10})]$.

Expérimentation de la méthode au Tableau des Entrées Intermédiaires du Sénégal

Pour son expérimentation, la méthode est appliquée au TEI du Sénégal. Pour cela nous nous proposons d'estimer, suivant trois étapes, le TEI de 2004 en prenant celui de 2003 comme référence.

Etape 1 : les seules contraintes sont la demande et l'offre de Consommations Intermédiaires du TEI de 2004.

Etape 2 : en plus des contraintes de ***l'étape 1*** les cellules nulles du TEI de 2003 sont stabilisées à zéro.

Etape 3 : en tenant toujours compte des contraintes des ***étapes 1 et 2***, certaines cellules sont fixées à leurs valeurs connues du TEI de 2004 en vue de simuler la possibilité de prendre en compte des informations supplémentaires dans le TEI.

Résultats

Vu la taille des matrices qui sont étudiées (42x42), nous allons commenter les résultats sur une matrice de rang (7x7) tirée du TEI principal. Les résultats de ***l'étape 3*** figurent, pour l'ensemble du TEI, à l'annexe C.

Tableau 1 :
Sous matrice du TEI à l'étape 1

	AVI	**AIN**	**ECH**	**SEF**	**PEC**	**AEX**	**TVP**	**S.TOTAL**
AVI	3 486	3	267	2	2	2	2	**3 764**
AIN	1	12 172	4 183	2	4	2	2	**16 366**
ECH	3 789	1 808	5 498	2	2	2	61 623	**72 724**
SEF	1	3	2	2	125	2	2	**137**
PEC	1	3	2	2	261	2	32 102	**32 373**
AEX	1	3	2	2	2	12	2	**24**
TVP	1	3	2	2	2	2	38	**50**
S.TOTAL	**7 282**	**13 993**	**9 955**	**15**	**398**	**23**	**93 771**	**125 438**

A ***l'étape 1*** (cf. tableau 1), la contrainte d'égalité des totaux marginaux est respectée. Cependant, les cellules qui sont, pour le TEI du Sénégal, naguère égales à zéro sont devenues non nulles. Ce qui veut dire que, contrairement à la réalité, tous les produits font l'objet de consommation intermédiaire.

Tableau 2 **:**
Sous matrice du TEI à l'étape 2

	AVI	**AIN**	**ECH**	**SEF**	**PEC**	**AEX**	**TVP**	**S.TOTAL**
AVI	3 495	-	267	-	-	-	-	**3 761**
AIN	-	12 198	4 180	-	4	-	-	**16 382**
ECH	3 801	1 814	5 502	-	-	-	61 656	**72 773**
SEF	-	-	-	-	126	-	-	**126**
PEC	-	-	-	-	263	-	32 173	**32 436**
AEX	-	-	-	-	-	12	-	**12**
TVP	-	-	-	-	-	-	38	**38**
S.TOTAL	**7 296**	**14 012**	**9 949**	**-**	**393**	**12**	**93 866**	**125 528**

A ***l'étape 2*** (cf. Tableau 2), la fixation des cellules nulles a permis de respecter la structure interne du TEI du Sénégal. Outre ce gain en précision, des évolutions sont notées au niveau des totaux marginaux de la sous matrice, bien que ceux du TEI restent les mêmes.

Tableau 3 **:**
Sous matrice des contraintes internes du TEI

	AVI	**AIN**	**ECH**	**SEF**	**PEC**	**AEX**	**TVP**
AVI	3 486	-	264	-	-	-	-
AIN	-	-	4 188	-	-	-	-
ECH	-	-	5 506	-	-	-	-
SEF	-	-	-	-	123	-	-
PEC	-	-	-	-	258	-	32 183
AEX	-	-	-	-	-	-	-
TVP	-	-	-	-	-	-	35

Tableau 4 **:**
Sous matrice du TEI à l'étape 3

	AVI	AIN	ECH	SEF	PEC	AEX	TVP	S.TOTAL
AVI	3 486	-	264	-	-	-	-	**3 750**
AIN	-	12 210	4 188	-	4	-	-	**16 402**
ECH	3 805	1 817	5 506	-	-	-	61 651	**72 780**
SEF	-	-	-	-	123	-	-	**123**
PEC	-	-	-	-	258	-	32 183	**32 441**
AEX	-	-	-	-	-	12	-	**12**
TVP	-	-	-	-	-	-	35	**35**
S.TOTAL	**7 291**	**14 027**	**9 958**	**-**	**385**	**12**	**93 869**	**125 543**

L'***étape 3*** (cf. tableau 3 et 4) met en exergue la particularité de la méthode MEC à savoir l'introduction de l'information supplémentaire collectée dans le cadre de l'actualisation du TEI. Ce qui permet de dépasser les limites de la méthode RAS qui n'offre pas la possibilité d'agir dans la structure interne de la matrice.

Néanmoins aussi bien pour la méthode MEC que pour la méthode RAS, les totaux marginaux sont respectés. La seule différence réside au niveau de la structure interne du TEI. Ce qui est illustré ci-dessous par les changements notés au niveau des totaux marginaux de la sous matrice.

Tableau 5 **:**
Comparaison totaux marginaux de la sous matrice avec les méthodes RAS et MEC

	OFFRE DE CI			DEMANDE DE CI		
	RAS	**MEC (étape 2)**	**ECART**	**RAS**	**MEC (étape 2)**	**ECART**
AVI	3 756	3 761	6	7 294	7 296	2
AIN	16 406	16 382	- 23	14 036	14 012	- 24
ECH	72 787	72 773	- 14	9 955	9 949	- 6
SEF	123	126	3	-	-	-
PEC	32 445	32 436	- 9	386	393	8
AEX	10	12	2	10	12	2
TVP	35	38	3	93 882	93 866	- 16
S.TOTAL	**125 562**	**125 528**	**- 34**	**125 562**	**125 528**	**- 34**
	OFFRE DE CI			**DEMANDE DE CI**		
	RAS	**MEC (étape 3)**	**ECART**	**RAS**	**MEC (étape 3)**	**ECART**
AVI	3 756	3 750	- 6	7 294	7 291	- 3
AIN	16 406	16 402	- 3	14 036	14 027	- 9
ECH	72 787	72 780	- 7	9 955	9 958	3
SEF	123	123	0	-	-	-
PEC	32 445	32 441	- 4	386	385	- 0
AEX	10	12	2	10	12	2
TVP	35	35	- 0	93 882	93 869	- 12
S.TOTAL	**125 562**	**125 543**	**- 19**	**125 562**	**125 543**	**- 19**

Conclusion

Cet article a montré l'applicabilité de la méthode de minimisation de l'Entropie Croisée pour l'actualisation du Tableau des Entrées Intermédiaires de la Comptabilité Nationale. La méthode la plus utilisée pour ce problème est la méthode RAS. Cependant elle présente des difficultés quant à la fixation de cellules ou le remplissage de cases vides pour la prise en compte des nouvelles informations structurelles. La méthode de l'Entropie Croisée, en comblant ce manquement, a l'avantage d'être simple d'utilisation et de demander peu d'information. Ainsi, en partant d'un Tableau de référence (TEI 2003 du Sénégal), l'application de la méthode a permis de générer un nouveau TEI respectant les contraintes liées aux nouvelles informations obtenues sur sa structure interne et ses totaux marginaux.

Recommandations

L'outil décrit dans l'article ci-dessus est une contribution aux techniques de projection des Tableaux des Entrées Intermédiaires (TEI).

Ainsi, il est, d'une façon générale, recommandé à tout problème lié au renseignement des données manquantes d'un tableau carré en connaissant l'ensemble ou une partie de ses totaux marginaux.

En particulier l'outil est recommandé :

- Pour les comptables nationaux qui, à défaut de disposer chaque année des données nécessaires à l'élaboration du TEI, ont besoin de l'actualiser en disposant de nouvelles informations sur la structure de l'économie étudiée. Pour son application il est conseillé de disposer d'un TEI de départ (pour l'année n-1), des cibles (pour l'année n) notamment la demande et l'offre de Consommations Intermédiaires (CI) et des éventuels changements à apporter à la structure de l'économie. De même, en disposant pour une année d'actualisation la consommation intermédiaire totale, un TEI de référence et uniquement une partie de l'offre ou de la demande de CI par produit ou par branche, il est toujours possible, par la méthode MEC, d'estimer un nouveau TEI.

- Pour la désagrégation des comptes de la MCS, notamment pour les agents institutionnels (Ménages, Etat, Entreprises, Reste du monde). Par exemple, l'outil permet d'estimer la structure interne de la matrice des transferts inter-ménages avec la connaissance du montant global des transferts reçus des ménages par les ménages.

Références Bibliographiques

Bahan D., Billodeau D., Lemelin A. Robichaud V. (2003), «Une matrice de comptabilité sociale birégionale pour le modèle d'équilibre général du ministère des Finances du Québec (MEGFQ)», ISBN 2-550-41623-6, Bibliothèque nationale du Québec.

Dubé J., Lemelin A. (2005), «Une application expérimentale de la méthode de minimisation de l'entropie croisée : l'estimation des flux d'échanges interrégionaux au Québec», CIRPEE, *Cahier de recherche* 05-25.

Dramani L., Laye O., Ndiaye D. (2007), «Estimation des coefficients techniques robustes pour l'économie sénégalaise», Agence Nationale de la Statistique et de la Démographie, Sénégal.

Fofana I., Lemelin A., Cockburn J. (2005), «Balancing a Social Accounting Matrix: Theory and Application», CIRPEE, Université de Laval.

Golan A., G. Judge et D. Miller (1996), «Maximum Entropy Economics: robust estimation with limited data», John Willey & Sons.

Jaynes, E.T. (1957), «Information theory and statistical mechanics», *Physical Review*, vol. 106, p.620-630 et vol. 108, p.171-190.

Kapur J.N et Kesavan H.K. (1992), «Entropy optimization principles with applications», Academic Press, Inc., San Diego, CA, 405 Pages.

Kulback S. (1959), «Information Theory and Statistics», John Willey & Sons Inc., Canada, 395 pages.

Robinson S., Cattaneo A., El-Said M. (2000), «Estimating a Social Accounting Matrix Using Cross Entropy Methods», IFPRI. Discussion Paper n° 58.

Robinson S., El-Said M. (2000), «GAMS Code for estimating a Social Accounting Matrix (SAM) using Cross Entropy (CE) Methods», IFPRI. Discussion Paper n° 64.

Robinson S., Cattaneo A. et El-Said M. (1998), «Estimating a Social Accounting Matrix Using Cross Entropy Methods», IFPRI. Discussion Paper n° 33.

Theil H. (1967), «Economics and information theory», Ran McNally & Company, Chicago, Studies in mathematical and managerial economics, 7, 488 pages.

Annexe A : Processus d'itérations de la méthode RAS

Soit A^0 la matrice de départ, la méthode consiste à générer une nouvelle matrice A telle que :

$$rA^0s = A$$

Sous la contrainte

$$\sum_i^n r_i = \sum_j^n s_j$$

Avec $r = (r_i)_{i\in\{1,\dots,n\}}$ et $s = (s_j)_{j\in\{1,\dots,n\}}$ respectivement les vecteurs totaux en ligne et en colonne de la nouvelle matrice A.

Une fois la contrainte respectée, le processus d'itération ci-dessous débute.

1. Si $A^0 = (x^0{}_{i,j})_{i,j\in\{1,\dots,n\}}$, on calcule, pour chaque ligne i, la somme des colonnes de A^0 de façon à obtenir un vecteur ligne m_i donnée par :

$$\forall\, i,\;\; m_i = \sum_j^n x^0{}_{i,j}$$

2. On calcule p_i pour chaque ligne i que l'on multiplie à chaque élément de la $i^{ème}$ ligne de la matrice A^0 :

$$p_i = \frac{r_i}{m_i} \quad et \quad x^1{}_{i,j} = p_i x^0{}_{i,j}$$

Ce qui donne la nouvelle matrice $A^1 = (x^1{}_{i,j})_{i,j\in\{1,\dots,n\}}$

3. Puis, pour chaque colonne j, on obtient le vecteur colonne n_j donnée par la somme des lignes de la matrice A^1 ainsi que de l'écart qu'il reste à combler par :

$$\forall\, j,\quad n_j = \sum_i^n x^1{}_{i,j} \quad et \quad ecart = \sum_J^n |n_j - s_j|$$

Si l'écart est plus grand que le seuil de convergence, on poursuit à l'étape 4.
Sinon, la matrice est équilibrée.

4. On calcule q_j pour chaque colonne j que l'on multiplie à chaque élément de la $j^{ème}$ colonne de la nouvelle matrice A^1 obtenue à l'étape 2 :

$$q_j = \frac{s_j}{n_j} \;\; et\; x^2{}_{i,j} = q_j x^1{}_{i,j}$$

Ce qui donne la nouvelle matrice $A^2 = (x^2{}_{i,j})_{i,j\in\{1,\dots,n\}}$

5. De nouveau le test de convergence est exécuté sur la nouvelle matrice A^2 en calculant respectivement le vecteur m_i et l'écart qu'il reste à combler :

$$\forall\, i,\;\; m_i = \sum_j^n x^2{}_{i,j} \;\; et \;\; ecart = \sum_i^n |m_i - r_i|$$

Si l'écart est inférieur au seuil de convergence, alors les itérations sont terminées, sinon on retourne à l'étape 1.

Annexe B : Code GAMS de la méthode de l'Entropie Croisée appliquée au TEI du Sénégal

```
$TITLE ENTROPIECROISEE

$STITLE Résolution du problème de minimisation de l'entropie croisée pour la projection du Tableau des Entrées Intermédiaires

*===============================CALIBRATION================================*

*------------------------Les ensembles utilises----------------------------*

SET I les branches d'activités
     /AVI    agriculture vivrière
      AIN    agriculture industrielle ou exportation
      ECH    élevage et chasse
      SEF    sylviculture exploitation forestière
      PEC    pêche
      AEX    activités extractives
      TVP    Transformation et conservation de viande poisson
      CGA    fabrication de corps gras alimentaires
      TGP    travail de grains fabrication de produits
      PAC    Fabrication de produits alimentaires céréaliers
      STR    fabrication de sucre transformation
      PAL    fabrication de produits alimentaires n.c.a
      BOI    fabrication de boissons
      PBT    fabrication de produits à base de tabac
      ECT    égrenage de coton et fabrication des textiles
      CUF    fabrication du cuir
      TBF    travail du bois et fabrication d'articles
      PCA    fabrication de papier carton
      RPC    raffinage pétrole cokéfaction
      PCH    fabrication de produits chimiques
      FCA    fabrication de produits en caoutchouc
      VEP    fabrication de verre poterie
      FMF    métallurgie fonderie
      MAC    fabrication de machines
      EAP    fabrication équipements appareils
      MTR    construction de matériels de transports
      MOB    fabrication de mobilier
      EGE    électricité gaz et eau
      BTP    construction
      COM    commerce
      SRE    services de la réparation
      SHR    services hébergement et restauration
      TRA    transports
      PTE    postes et télécommunications
      SFI    services financiers
      AIM    activités immobilières
      AEN    activités des services aux entreprises
      APU    activités administration publique
      EDF    éducation et formation
      ASS    activités de santé et action sociale
      ACP    activités à caractère collectif ou personnel
      SIF    Services intermédiation financière i. m.
      CT     correction territoriale /
```

```
ALIAS (I, J);

*--------------------------------Liste des variables--------------------------*

PARAMETER

TEIO(I,J)            Matrice des entrées intermédiaires du Sénégal
TEI_CONT(I,J)        Matrice des contraintes à l'intérieur du TEI
TLO(J)               Totaux marginaux des lignes
TCO(I)               Totaux marginaux des colonnes
ECHTOTO              Somme des totaux marginaux
QIJO(I,J)            Matrice de la distribution de probabilité à priori
QPTJO(I)             Totaux marginaux des lignes en distribution de probabilité à priori
QIPTO(J)             Totaux marginaux des colonnes en distribution de probabilité à priori
OFFCI(I)             Offre de CI de la branche I
OFFTOTO              Offre totale
DEMCI(J)             Demande de CI de la branche J
DEMTOTO              Demande totale
PIPTO(I)             Totaux marginaux des lignes en distribution de probabilité
PPTJO(J)             Totaux marginaux des colonnes en distribution de probabilité
EPSILON              Empêche des divisions par zéro

*Les données sont incluses dans les fichiers suivants :
*Fichier contenant la matrice d'information à priori
$INCLUDE DATATEI.prn
*$OFFDELIM
*Fichier contenant les contraintes à l'intérieur du TEI
$INCLUDE DATATEI_cont.prn
*Fichier contenant les totaux marginaux des lignes et colonnes
$INCLUDE CONT_DEM.prn
$INCLUDE CONT_OFF.prn

*------------------------Assignation des valeurs de base---------------------*

TEIO(I,J)          =     TEISN(I,J);
TEI_CONT(I,J)      =     TEISN_CONT(I,J);
TLO(J)             =     SUM(I,TEIO(I,J));
TCO(I)             =     SUM(J,TEIO(I,J));
ECHTOTO            =     SUM((I,J),TEIO(I,J));

QIJO(I,J)          =     TEIO(I,J)/ECHTOTO;
QPTJO(I)           =     TCO(I)/ECHTOTO;
QIPTO(J)           =     TLO(J)/ECHTOTO;
OFFCI(I)           =     OFFRE(I,"OFFCI");
OFFTOTO            =     SUM(I,OFFCI(I));
DEMCI(J)           =     DEMANDE("DEMCI",J);
DEMTOTO            =     SUM(J,DEMCI(J));

*Données servant à la construction des contraintes

PIPTO(I)           =     OFFCI(I)/OFFTOTO;
PPTJO(J)           =     DEMCI(J)/DEMTOTO;
EPSILON            =     0.000001;
```

```
*-------------------------Nom des variables utilisées------------------------*

VARIABLE

TEI(I,J)                Matrice des entrées intermédiaires du Sénégal
QIJ(I,J)                Matrice de la distribution de probabilité a priori
QPTJ(J)                 Totaux marginaux des lignes en distribution de probabilité a priori
QIPT(I)                 Totaux marginaux des colonnes en distribution de probabilité a priori
PIJ(I,J)                Matrice des probabilités à posteriori
PIPT(I)                 Totaux marginaux des lignes en distribution de probabilité
PPTJ(J)                 Totaux marginaux des colonnes en distribution de probabilité
OBJ                     Fonction objectif pour chaque produit

*--------------------------Assignation des valeurs--------------------*

*L'assignation des valeurs se fait dans le même bloc que les équations pour chaque produit.

*-----------------------------Les équations-----------------------------------*

EQUATIONS

EQCTM(J)                   Equation de contrainte de respect des tot. margin. (colonnes)
EQLTM(I)                Equation de contrainte de respect des tot. margin. (lignes)
EQSE                    Contrainte des éléments à donner un une fois additionnes
EQCE                    Equation de minimisation de l'entropie croisée

*Assignation de la matrice de départ

TEI.fx(I,J)                =     TEIO(I,J)+TEI_CONT(I,J);
QIJ.FX(I,J)                =     QIJO(I,J);
PIJ.LO(I,J)                =     0;
PIJ.UP(I,J)                =     1;
PPTJ.FX(J)                 =     PPTJO(J);
PIPT.FX(I)                 =     PIPTO(I);
OBJ.LO                     =     0;

*Définition de l’ensemble des cellules nulles pour le TEI de départ

SET zeros(i,j);
zeros(I,J)$(QIJ.l(I,J) eq 0)     = YES;

*Fixation des cellules nulles

PIJ.LO(I,J)$zeros(I,J)        =     0;
PIJ.UP(I,J)$zeros(I,J)        =     1e-12;

Définition de l’ensemble des cellules devant contenir les contraintes sur la structure interne du TEI

SET nzeros(i,j);
nzeros(I,J)$(TEI_CONT(I,J) ne 0)     = YES;

*Fixation des contraintes de la structure interne du TEI

PIJ.FX(I,J)$nzeros(I,J)        =     TEI.L(I,J)/OFFTOTO;

*Ecriture des équations

EQCTM(J)..               SUM(I,PIJ(I,J))          =E=     PPTJ(J);
EQLTM(I)..               SUM(J,PIJ(I,J))          =E=     PIPT(I);
EQSE..          SUM((I,J),PIJ(I,J))         =E=     1;
EQCE..          OBJ                         =E=     SUM((I,J),(PIJ(I,J)*log(PIJ(I,J)+EPSILON)-
                                                         PIJ(I,J)*log(QIJ(I,J)+EPSILON))) ;
```

```
*-------------------Résolution du problème ----------------------------*

OPTION ITERLIM=100000;
OPTION DECIMALS=8;
MODEL ENTROPIECROISEE /ALL/;
OPTION NLP=MINOS5;
SOLVE ENTROPIECROISEE USING NLP MINIMIZING OBJ
```